LIM
Version 0.7
User's Guide

December 2025

Synclesis, Inc.

Urbana, IL

This version is the first major release of LIM. The development of LIM and its algorithms is an ongoing effort at Synclesis, and therefore not all of the intended capabilities of the simulator have been implemented in full yet.

1. INTRODUCTION

LIM is a general-purpose circuit simulation program. The current version focuses on nonlinear transient analysis. It will also predict DC behavior. Circuits may contain resistors, capacitors, inductors, mutual inductors, independent voltage and current sources. Four types of dependent sources are also supported. Systems with frequency-dependent behavior can be handled with controllable accuracy. This version supports lossy and lossless single or coupled transmission lines and multi-port macromodels described by their frequency-dependent network parameters.

LIM currently supports three of most common semiconductor devices: diodes, BJTs, and MOSFETs. The development of models for these devices is a work in progress. Current models supported are the Ebers-Moll model for BJTs and level 1for MOSFETs.

1.1. TYPES OF ANALYSIS

1.1.1. Transient Analysis

The transient analysis in LIM is performed using the latency insertion algorithm in which voltage and current variables are computed in a leapfrog manner as a function of time. The same general algorithm is applied to all elements in a circuit

The time interval is specified on a .TRAN control line.

1.3. STABILITY

Although the algorithm used in LIM has been found to be reliable, in some cases it is not stable and leads to a diverging solution. When this failure occurs, erroneous values are produced.

Failure to achieve a stable solution can be caused by incorrect circuit connections or a time step that is too large. LIM uses a fictitious capacitor to ground for each node and a fictitious series inductor for some branches. Values for these fictitious elements can affect the stability, accuracy and speed of the simulation. The .LIM control line is used to control these parameters.

1.4. RUNNING LIM

The LIM simulation engine is invoked by issuing the command:

```
lim   netlist  directory  index
```

Where netlist is the file name of the netlist containing the circuit description. The file name should include the full path from the current directory to the file of interest. directory is the name of the directory in which the output files resulting from the run will be stored.

Within the directory several files are created:

"*out_file*" which is a run summary indicating all the key properties of the network and of the simulation.

Symbol	**Description**
Nnodes	Total number of nodes
dtUser	Time step
dtMax	Maximum allowable time step
Npts	Number of points
Lsmal	Fictitious inductor
Csmal	Fictitious capacitor

"*data_file*" which tabulates the results of the transient simulation and list the time in one column and the voltages in subsequent column in the order in which they appear in the .print statement in the netlist.

"*lim_log*" a log file that contains some of the key parameters from the simulation.

"*netfile*" which contains the netlist with all the nodes renumbered.

index is a digit that can be 1 or 0

If index=0, only the set up is performed without simulation. Program halts before simulation. The summary of the setup is written in the file *out_file*.

If index=1, the complete simulation is performed.

2. NETLIST DESCRIPTION

2.1. General Structure and Conventions

LIM follows a nomenclature that is closely related to that of SPICE, version 3. Most of the format, conventions and netlist representations are the same as SPICE3.

The circuit to be analyzed is represented to LIM by a set of element lines describing the circuit and element values, and a set of control lines, that define the model parameters and the simulation parameters.

Lines starting with a "*" are comment lines and are ignored. The order in which the lines are listed is arbitrary and the last line in the netlist must be ".END". Continuation lines start with a "+" .

Each element in the circuit must be described by a line. Each element line contains the element name, the circuit nodes to which the element is connected, and the values of the parameters that determine the electrical characteristics of the element. The LIM element line starts with an alphanumeric string representing the element name and in which the first letter specifies the element type is denoted by with an alphanumeric string with the first letter of the element name specifying the element type. For example, a capacitor name must begin with the letter C and can contain one or more characters. As an example, C, C1, CIN, CLOAD, and C192A are valid capacitor names. Details of each type of device are supplied in a following section.

Items on a line are separated by one or more blanks. A comma, an equal ('=') sign, or a left or right parenthesis; extra spaces are ignored. A line may be continued by entering a '+' (plus) in column 1 of the following line; LIM continues reading beginning with column 2.

A name item must begin with a letter (A through Z) and cannot contain any delimiters. A number may be an integer, a floating point. Exponent format (1e-14, 2.65e3) is also allowed.

Node names may be arbitrary character strings. The reference (ground) node must be named '0'.

2.2 DC Analysis

The dc solution in LIM is taken as the steady-state response resulting from all independent DC sources switching as steps (or ramps) to their respective values. It is in essence a solution to a transient simulation in which all variables eventually converge to constant values. In special cases involving oscillators, a constant DC value will not be reached; the DC analysis simulation will instead produce oscillations at a specific frequency.

2.3 Control Lines

2.3.1 .OP Line

The .OP line is used to control the parameters of the DC analysis. It indicates that only the DC analysis should be performed for the circuit. Execution will stop once the DC solution for the network has been reached.

General form:

```
.OP TSTEP TSTOP
```

The DC operating point analysis in LIM is performed through a transient analysis. TSTEP is the is the time step used for that transient analysis. TSTOP is a time limit at which to stop the transient simulation. It is anticipated that at TSTOP, all the voltage and current values will converge to their final value which represent the DC operating point. For this analysis, all values start at zero. The DC independent sources are next ramped up from zero until they reached their DC values. After enough time has elapsed.

Examples:

```
.OP 1E-06 1E-09
```

2.3.2 .TRAN Line

The .TRAN line is used to control the parameters of the transient simulation

General form:

```
.TRAN TSTEP TSTOP
```

TSTEP is the is the time step used for the simulations. In this version of LIM, the time step is constant. TSTOP is the duration of the simulation. The ratio TSTOP/TSTEP determines the number of transient simulation points.

Examples:

```
.TRAN 1E-06 1E-09
```

2.3.3 .PRINT Line

The .PRINT line is used to print the output of the simulation

General form:

```
.PRINT PRTYPE V(N1) V(N2) V(N3) V(N4)
.PRINT PRTYPE I(EL1) I(EL2) I(EL3) I(EL4)
```

The .PRINT line defines the contents of a tabular listing of one to four output variables. PRTYPE is the format of the file. The different formats are as follows:

- ASC: Regular ASCII file format.
- BIN: Binary file format.
- CSV: CSV file format.

Note: the "TRAN" file format is still accepted and will be interpreted as "ASC". The variables to be printed can be voltages or currents.

- In the case of voltage, the node string value is specified between the parentheses.
- In the case of current, the name of the element appears between the parentheses. Some restrictions apply. The element must be a two-terminal element. Currently, resistors, capacitors, inductors, diodes, voltage and current sources (dependent or independent) are allowed. The direction of the current is taken from the first to the second of the two-terminal element as specified on the netlist.

Examples:

```
.PRINT BIN V(4) V(3)
.PRINT ASC I(R4) I(L3)
```

2.3.4 .PLOT Lines

The .PRINT line is used to print the output of the simulation

General form:

```
.PLOT PRTYPE V(N1) V(N2) V(N3) V(N4)
.PLOT PRTYPE I(EL1) I(EL2) I(EL3) I(EL4)
```

The .PLOT line defines the contents of a tabular listing of one to four output variables. PRTYPE is the type of analysis. For this version, only TRAN is allowed. The variables to be plotted can be voltage or current.

- In the case of voltage, the node string value is specified between the parentheses.
- In the case of current, the name of the element appears between the parentheses. Some restrictions apply. The element must be a two-terminal element. Currently, resistors, capacitors, inductors, diodes, voltage and current sources (dependent or independent) are allowed. The direction of the current is taken from the first to the second of the two-terminal element as specified on the netlist.

Examples:

```
.PLOT TRAN V(4) V(3)
.PLOT TRAN I(R4) I(L3)
```

2.3.5 .LIM Line

Several parameters of the simulations available in LIM can be altered to control the accuracy, speed and stability. These parameters may be changed via the .LIM command.

General form

```
.LIM R=value G=value L= value C=value...
```

Each parameter refers to a LIM variable that controls one or several simulation constraints. They are defined below

C: Value of fictitious capacitance placed at each node of the network. Nodes that already have a capacitor to ground will ignore the fictitious capacitance. In general, it is best to choose a value for C that is at least one order of magnitude smaller than the smallest capacitance in the network. As C decreases, accuracy increases for a given time step at the cost of stability.

L: Value of fictitious inductance placed at each branch of the network. Branches that are inductors will ignore the fictitious inductance. In general, it is best to choose a value for L that is at least one order of magnitude smaller than the smallest inductance in the network. As L decreases, accuracy increases for a given time step at the cost of stability.

G: Value of fictitious conductance placed at each node of the network. Nodes that already have a resistor to ground will ignore that fictitious conductance. In general, G is chosen to be very small. In some rare cases, a relatively larger value of G it may help facilitate the simulations at the cost of accuracy. This option is added to provide more flexibility in the simulations.

P: Frequency at which simulation variables that are to be printed will be printed.

W: Frequency at which simulation variables will be written into data file.

O: Frequency at which simulation variables that are to be plotted will be plotted.

D: DC analysis control parameter. A value of 0 means that the DC analysis will be performed before the transient analysis. A value of 1 indicates that only the DC analysis will be performed.

S: Slew rate control parameter for DC analysis. The value assigned to S will be multiplied by the time step and used as the rise time of the step voltages and step currents for the DC analysis.

Example

```
.LIM C=1.010e-14 L=0.1e-09 G=1.0e-20 P=80 O=10 D=0
```

2.3.6 .MODEL Line - Device Models

The .MODEL line contains parameter values for special devices. For these device types, each device element line contains the device name, the nodes to which the device is connected, and the device model name.

General form:

```
.MODEL MNAME TYPE(PNAME1=PVAL1 PNAME2=PVAL2 ... )
```

Example:

```
.MODEL MOD1 NPN (BF=50 IS=1E-13 VBF=50)
```

The .MODEL line specifies the parameter values for that particular element line. Several different elements may refer to the same model line with the same parameters. In that case, they must use the same model name (MNAME).

MNAME in the above is the model name. Model types currently available in this version are:

- LTRA: Lossy transmission line model
- D: Diode model
- NPN: NPN bipolar junction transistor model
- PNP: PNP bipolar junction transistor model
- NMOS: N-channel MOSFET model
- PMOS: P-channel MOSFET model

Parameter values are defined by appending the parameter name followed by an equal sign and the parameter value. Model parameters that are not given a value are assigned default values that depend on the model type. Model parameters and default values are listed in the respective sections along with the description of device element lines.

2.3.7 .SUBCKT Line - Subcircuits

A subcircuit is defined as a grouping of elements. A set of nodes is next defined to provide access to the subcircuit from external elements or other subcircuits. The group of elements within a subcircuit can be accessed wherever the subcircuit is referenced. There is no limit on the size or complexity of subcircuits, and subcircuits may contain other subcircuits.

General form:

```
.SUBCKT subcktname N1 N2 N3 ...
```

A circuit definition starts with a .SUBCKT line. The name of the subcircuit is SUBCKTNAME. N1, N2, ..are the external nodes; they cannot be zero. The element lines that follow represent the subcircuit. The last line is .ENDS . Control lines cannot be used within a subcircuit definition. Device model definitions can be invoked in a subcircuit block. In that case, they are strictly local and cannot be known outside of the subcircuit block. Except for the external nodes specified in the .SUBCKT line, all nodes are internal to the subcircuit block except for 0 (ground reference node) which is always a global node.

Example:

```
.SUBCKT OPAMP 1 2 3 4
```

Subcircuit Calls – X-element

To invoke a subcircuit in a LIM netlist, the X element is used. This consist of an element with an alphanumeric name that starts with X followed by the circuit nodes and the subcircuit name. The circuit nodes will map according to the order in which they were defined in the `.SUBCKT` definition for the subcircuit `SUBCKTNAME`.

General form:

```
XYZA N1 N2 N3 ... SUBCKTNAME
```

Example:

```
X1 2 4 17 3 1 MULTI
```

2.3.8 .MACR Line

The .MACR line is a special control line used to perform model-order reduction analysis on multi-port S-parameter data files. The .MACR command will override all other commands in the netlist and cause execution to terminate after the model-order reduction analysis.

General form:

```
.MACR
```

LIM will carry out the pole-residue extraction for all the N-elements of the netlist and exit right after without proceeding to the simulation steps.

Example:

```
.MACR
```

2.3.9 .PARAM Line

The .PARAM line is a special control line used to assign numerical values to identifiers. More than one assignment per line is possible using a space as the separator. Parameter identifier names must begin with an alphabetic character.

General form:

```
.PARAM <ident>=<val> <ident1>=<val1>
```

where <ident> is a string and <val> is the numerical value assigned to the variable <ident>. The parser will assign the value to all instances of the string within the netlist.

Examples:

```
.PARAM CVAL=10p LVAL=12u
```

3. CIRCUIT ELEMENTS AND MODELS

LIM is a circuit simulator that establishes the relationship between voltage and current variables through circuit elements. LIM distinguishes two types of elements. Simple elements and devices. Simple elements are two terminal constructs that define a branch. By convention, positive current flow is in the direction of voltage drop. Following is a sequence of simple elements.

1. **Resistors - R- Element**

General form:

```
RXYZ N1 N2 VALUE
```

Examples:

```
R1 1 2 100
RC1 12 17 1E+03
```

N1 and N2 are the two element nodes. VALUE is the resistance (in ohms).

2. Capacitors – C- Element



General form:

```
CXYZ N+ N- VALUE
```

Examples:

```
CBYP 13 0 1UF
COSC 17 23 10U IC=3V
```

N+ and N- are the positive and negative element nodes, respectively. VALUE is the capacitance in Farads.

3. Inductors – L-Element

General form:

```
LYZX N+ N- VALUE
```

Examples:

```
LLINK 42 69 1UH
LSHUNT 23 51 10U IC=15.7MA
```

N+ and N- are the positive and negative element nodes, respectively. VALUE is the inductance in Henries.

4. Coupled (Mutual) Inductors – K-Element



General form:

```
KXYZ LYZX LABC VALUE
```

Examples:

```
K43 LAA LBB 0.999
KXFRMR L1 L2 0.87
```

`LYZX` and `LABC` are the names of the two coupled inductors, and VALUE is the coefficient of coupling, K, which must be greater than 0 and less than or equal to 1. Using the 'dot' convention, place a 'dot' on the first node of each inductor.

5. Independent voltage sources – V-Element

DC Voltage Source:

General Form:

```
VXYZ N+ N- DC VALUE
```

N+ and N- are the positive and negative nodes, respectively. Note that voltage sources need not be grounded. DC indicates that it is a DC source. Value is the voltage in volts.

Example:

```
VCC 10 0 DC 6
```

Pulse Voltage Source

General Form:

```
VXYZ N+ N- PULSE(V1 V2 TD TR TF PW PER)
```

N+ and N- are the positive and negative nodes, respectively. PULSE indicates that it is a pulsed source. The pulse parameters are.

Parameter	Unit
V1 (initial value)	Volts
V2 (pulsed value)	Volts
TD (delay time)	seconds
TR (rise time)	seconds
TF (fall time)	seconds
PW (pulse width)	Seconds
PER (period)	seconds

Example:

```
VIN 3 0 PULSE(-1 1 2NS 2NS 2NS 50NS 100NS)
```

Sinusoidal Voltage Source

General Form:

```
VXYZ N+ N- SIN(VO VA FREQ TD THETA)
```

The signal has a sinusoidal shape with time. SIN indicates that it is a sinusoidal source. The algorithm follows the function:

$$V_g = V_o + V_A e^{-\theta(t-t_d)} \sin\left[2\pi f\left(t-t_d\right)\right]$$

or

```
VG = VO+VA*exp(-THETA*(t-TD)*sin(2*PI*FREQ*(t-TD))
```

The sine parameters are

Parameter	**Unit**
VO (offset)	Volts
VA (amplitude)	Volts
FREQ (frequency)	Hz
TD (delay) 0.0	seconds
THETA (damping factor)	1/second

Example:

```
VIN 3 0 SIN(0 1 100MEG 1NS 1E10)
```



Exponential Voltage Source

General Form:

```
VXYZ N+ N- EXP(V1 V2 TD1 TAU1 TD2 TAU2)
```

EXP indicates that it is an exponential source. The exponential algorithm follows the function:

$$V_g = \begin{cases} V_1, & for\ \ 0 < t < t_{d1} \\ V_1 + \left(V_2 - V_1\right)\left[1 - e^{-\frac{(t-t_{d1})}{\tau_1}}\right], & for\ \ t_{d1} < t < t_{d2} \\ V_1 + \left(V_2 - V_1\right)\left[1 - e^{-\frac{(t-t_{d1})}{\tau_1}}\right] + \left(V_1 - V_2\right)\left[1 - e^{-\frac{(t-t_{d2})}{\tau_2}}\right], & for\ t > t_{d2} \end{cases}$$

The parameters are listed below

Parameter	**Unit**
V1 (initial value, V_1)	Volts
V2 (pulsed value, V_2)	Volts
TD1 (rise delay time, t_{d1})	seconds

TAU1 (rise time constant, τ_1)	seconds
TD2 (fall delay time, t_{d2})	second
TAU2 (fall time constant, τ_2)	seconds

Example:

```
VIN 3 0 EXP(-4 -1 2.0E-9 30.0E-9 60.0E-9 40.0E-9)
```

Gaussian Voltage Source

General Form:

```
VXYZ N+ N- GAUS(V1 TD TAU)
```

GAUS indicates that it is a Gaussian source. The Gaussian algorithm follows the function:

$$V_g = V_1 e^{-\frac{(t-t_d)^2}{2\tau^2}}$$

The parameters are listed below

Parameter	**Unit**
V1 (magnitude value, V_1)	Volts
TD (time delay, t_d)	seconds
TAU (time constant, τ)	seconds

Example:

```
VIN 3 0 GAUS(1.0 2.0E-09 8.0E-09)
```

Surge Voltage Source



General Form:

```
VXYZ N+ N- SURG(V0 K TD A B)
```

SURG indicates that it is a voltage surge. The surge algorithm follows the equation:

$$V_g = \begin{cases} 0, & \text{for } t \le t_d \\ V_o k\left(e^{-a(t-t_d)} - e^{-b(t-t_d)}\right), & \text{for } t \ge t_d \end{cases}$$

The parameters are listed below

Parameter	**Unit**
V0 (initial value, E_o)	Volt
K (attenuation factor, k)	-
TD (time delay)	second
A (rise time constant, a)	$(\text{second})^{-1}$
B (fall time constant, b)	$(\text{second})^{-1}$

Examples:

```
VIN 3 0 SURG(70  1.3 1.0e-09  4.0e+07 6.0e+08)
```

Piece-Wise Linear Voltage Source

General Form:

```
VXYZ N+ N- PWL(T1 V1 T2 V2 ... TK VK ... TN VN)
```

At each time Ti, the voltage at the source is Vi. Intermediate time and voltage points are obtained via interpolation.

Example:

```
VCK 7 5 PWL(0 -7 10.0e-9 -7 11.0e-9 -3 17.0e-9 -3
18.0e-9 -7 50.0e-9 -7)
```

User Defined Voltage Source

General Form:

```
VXYZ N+ N- USERDEF Filename
```

This form indicates a piecewise linear description of the voltage source can be read from a file. The points t_i and $v(t_i)$are read directly from the file `Filename` in the format:

t_1 $v(t_1)$
t_2 $v(t_2)$
t_3 $v(t_3)$

The units are seconds and volts. During a run, if the time specified is larger than the last time point, then the voltage value corresponding to the last time point is returned.

Example:

```
VP 1 0  USERDEF voltage_listing.txt
```

Bit Sequence Voltage Source

General Form:

```
VXYZ N+ N- BITSEQ(V1 V2 XXXYYY BR TR TF)
```

This form indicates a bit sequence of zeros and ones. The sequence repeats itself. The parameters are listed below

Parameter	**Unit**
`V1` (initial value,)	volts
`V2` (Final value)	volts
`XXXYYY`	Bit Sequence
`BR` (bit rate)	Bits/seconds
`TR` (rise time)	seconds
`TF` (fall time)	second

Example:

```
VP 1 0  BITSEQ (0.33 1.8  0110011 1.0e+09 0.1n 0.1n)
```

Pseudo-Random Bit Sequence Voltage Source

General Form:

```
VXYZ N+ N- PRBSK(V1 V2 BR TR TF)
```

This form indicates a bit sequence of zeros and ones that is randomly generated. **`K`** is an integer associated with the number of random bits in a repeated pattern. Currently, **`K`** =3, 4, 5, 7, 15, 31 are supported.

The number of bits in a pattern is given by:

Nbits = $2^K - 1$

The parameters are listed below

Parameter	**Unit**
`V1` (initial value,)	volts
`V2` (Final value)	volts
`BR` (bit rate)	bits/seconds
`TR` (rise time)	seconds
`TF` (fall time)	seconds

Example:

```
VP 1 0  PRBS5 (0.0 3.3 1.0e+09  1.0e-10  1.0e-10 )
```

6. Independent current sources – I-Element

DC Current Source:

General Form:

```
IXYZ N+ N- DC VALUE
```

N+ and N- are the positive and negative nodes, respectively. Positive current is assumed to flow from the positive node, through the source, to the negative node. A current source of positive value forces current to flow out of the N+ node, through the source, and into the N- node. DC indicates that it is a DC source. Value is the current in amperes.

Examples:

```
ICC 10 0 DC 6
```

Pulsed Current Source

General Form:

```
IXYZ N+ N- PULSE(I1 I2 TD TR TF PW PER)
```

Same convention applies for the parameters as in the pulse voltage source case with voltage in volts replaced by current in amperes.

Examples

```
IIN 3 0 PULSE(-1 1 2NS 2NS 2NS 50NS 100NS)
```

Sinusoidal Current Source

General Form:

```
IXYZ N+ N- SIN(IO IA FREQ TD THETA)
```

Same convention applies for the parameters as in the sinusoidal voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 SIN(0 1 100MEG 1NS 1E10)
```

Exponential Current Source

General Form:

```
IXYZ N+ N- EXP(I1 I2 TD1 TAU1 TD2 TAU2)
```

Same convention applies for the parameters as in the exponential voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 EXP(-4 -1 2.0E-9 30.0E-9 60.0E-9 40.0E-9)
```

Gaussian Current Source

General Form:

```
IXYZ N+ N- GAUS(I1 TD TAU)
```

Same convention applies for the parameters as in the Gaussian voltage source case with voltage in volts replaced by current in amperes.

Example:

```
IIN 3 0 GAUS(1.0 2.0E-09 8.0E-09)
```

Surge Current Source

General Form:

```
IXYZ N+ N- SURG (I0 K TD A B)
```

Same convention applies for the parameters as in the exponential voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 SURG (70  1.3 1.0e-09  4.0e+07 6.0e+08)
```

Piecewise Linear Current Source

General Form:

```
IXYZ N+ N- PWL(T1 I1 T2 I2 ... TK IK ... TN IN)
```

Same convention applies for the parameters as in the piecewise linear voltage source case with voltage in volts replaced by current in amperes.

Example:

```
ICK 7 5 PWL(0 -7 10.0e-9 -7 11.0e-9 -3 17.0e-9 -3
18.0e-9 -7 50.0e-9 -7)
```

Bit Sequence Current Source

General Form:

```
IXYZ N+ N- BITSEQ(I1 I2 XXXYYY BR TR TF)
```

This form indicates a bit sequence of zeros and ones. The sequence repeats itself. The parameters are listed below

Parameter	**Unit**
`I1` (initial value,)	Amps
`I2` (Final value)	Amps
`XXXYYY`	Bit Sequence
`BR` (bit rate)	Bits/seconds
`TR` (rise time)	seconds
`TF` (fall time)	second

Example:

```
IP 1 0  BITSEQ (0.33 1.8  0110011 1.0e+09 0.1n 0.1n)
```

Pseudo-Random Bit Sequence Current Source

General Form:

```
IXYZ N+ N- PRBSK(I1 I2 BR TR TF)
```

This form indicates a bit sequence of zeros and ones that is randomly generated. **`K`** is an integer associated with the number of random bits in a repeated pattern. Currently, **`K`** =3, 4, 5, 7, 15, 31 are supported.

The number of bits in a pattern is given by:

Nbits = $2^K - 1$

The parameters are listed below

Parameter	**Unit**
`I1` (initial value,)	amps
`I2` (Final value)	amps
`BR` (bit rate)	bits/seconds
`TR` (rise time)	seconds
`TF` (fall time)	seconds

Example:

```
IP 1 0  PRBS5 (0.0 1.8m 1.0e+09  1.0e-10  1.0e-10 )
```

User Defined current Source

General Form:

`IXYZ N+ N- USERDEF Filename`

This form indicates a piecewise linear description of the voltage source can be read from a file. The points t_i and $i(t_i)$are read directly from the file `Filename` in the format:

```
t1    i(t1)
t2    i(t2)
t3    i(t3)
```

The units are seconds and amperes. During a run, if the time specified is larger than the last time point, then the current value corresponding to the last time point is returned.

Example:

`IP 1 0  USERDEF current_listing.txt`

7. Linear Voltage-Controlled Current Sources – G-Element



General form:

`GXYZ N+ N- NC+ NC- VALUE`

Example:

`G1 2 0 5 0 0.1MMHO`

N+ and N- are the positive and negative nodes, respectively. Current flow is from the positive node, through the source, to the negative node. NC+ and NC- are the positive and negative controlling nodes, respectively. VALUE is the transconductance (in mhos).

8. Linear Voltage-Controlled Voltage Sources – E-Element



General form:

`EXYZ N+ N- NC+ NC- VALUE`

Example:

`E1 2 3 14 1 2.0`

N+ is the positive node, and N- is the negative node. NC+ and NC- are the positive and negative controlling nodes, respectively. VALUE is the voltage gain.

9. Linear Current-Controlled Current Sources – F-Element



General form:

```
FXYZ N+ N- VNAM VALUE
```

N+ and N- are the positive and negative nodes, respectively. Current flow is from the positive node, through the source, to the negative node. VNAM is the name of a voltage source through which the controlling current flows. The direction of positive controlling current flow is from the positive node, through the source, to the negative node of VNAM. VALUE is the current gain

Example:

```
F1 13 5 VSENS 5
```

10. Linear Current-Controlled Voltage Sources – H-Element



General form:

```
HXYZ N+ N- VNAM VALUE
```

N+ and N- are the positive and negative nodes, respectively. VNAM is the name of a voltage source through which the controlling current flows. The direction of positive controlling current flow is from the positive node, through the source, to the negative node of `VNAM`. `VALUE` is the transresistance (in ohms).

Examples:

```
HX 5 17 VZ 0.5K
```

11. Lossless Transmission Lines – T-Element



General Form

```
TXYZ N1 N1REF N2 N2REF ZO=VALUE TD=VALUE
```

N1 and N1REF are the signal and reference nodes at the near end (port 1) of the line; N2 and N2REF are the signal and reference nodes at the far end (port 2) of the line. `ZO` is the characteristic impedance. TD is the transmission delay.

Example:

```
T1 1 0 2 0 ZO=50 TD=10NS
```

12. Lossy Transmission Lines – O-Element



General form:

```
OXYZ N1 N1REF N2 N2REF RLGCFILENAME LEN=VAL
```

`N1` and `N1REF` are the signal and reference nodes at the near end (port 1) of the line; `N2` and `N2REF` are the signal and reference nodes at the far end (port 2) of the line `RLGCFILENAME` is the name of the file containing the R, L, G and C parameters as a function of frequency. `LEN` is the length of the line in meters.

Examples:

```
O23 1 0 2 0 LOSSYMOD LEN=0.3
OCONNECT 10 5 20 5 INTERCONNECT LEN=0.1
```

Format of RLGC file

The RLGC file containing the data for the transmission line is arranged as follows

```
Number of frequency points (M)
Frequency-factor R_factor L_factor G_factor C_factor
f1 R(f1)
f2 R(f2)
:
fk R(fk)
:
fM R(fM)
f1 L(f1)
f2 L(f2)
:
fk L(fk)
:
fM L(fM)
f1 G(f1)
f2 G(f2)
:
fk G(fk)
:
fM G(fM)
f1 C(f1)
f2 C(f2)
```

```
:
fk C(fk)
:
fM C(fM)
```

Actual values are obtained by multiplying the parameters by their respective factors. The parameters `Frequency-factor` `R_factor L_factor G_factor C_factor` refer to the scale of the units in which the parameters are read.

13. Coupled Lossless Muticonductor Lines - U element

In LIM, the lossless multiconductor line is represented by the U-element. This consists a a group of M coupled transmission lines that have constant inductance and capacitance (self and mutual) parameters that do not change with frequency.



General form:

```
UXYZ N=M NN1 NN2 … NNM NNREF NF1 NF2 .. NFM NFREF UELFILENAME
LEN=LENGTH
```

`N=M` is the number of coupled lines. `NN1, NN2`,.. are the signal nodes at the near end of the multiconductor system. `NNREF` is the reference node at the near end of the multiconductor system. `NF1, NF2`,. are the signal nodes at at the far end of the multiconductor system. NFREF is the reference node at the far end of the multiconductor system. `UELFILENAME` is the name of the file containing the inductance and capacitance matrices [L] and [C]. These are M by M matrices that contains the Maxwellian elements. `LEN` is the length of the lines.

Examples:

```
U1 N=6 2 3 4 5 6 7 0 8 9 10 11 12 13 0 LCFILE LEN=14
```

Format of LC file

The LC file containing the data for the multiconductor system contains the Maxwellian inductance and capacitance lower-left triangular matrix elements.

```
+ Number of lines (N)
+ INFINITY
```

```
+ L11
+ L21
+ L22
+ …
+…
+ LNN
+ INFINITY
+ C11
+ C21
+ C22
+ …
+…
+ CNN
```

The line "+INFINITY" is to indicate that the inductance and capacitance values correspond to those for the infinite frequency limit.

14. Coupled Lossy Muticonductor Lines - W element

The W element is used to represent coupled transmission lines with frequency-dependent parameters. It can be used to simulate both lossy and lossless lines.



General form:

```
WXYZ N=Numlines NN1 NN2 … NNM NNREF NF1 NF2 .. NFM NFREF
WELFILENAME  LEN=VAL.
```

`N` is the number of lines, `N1a, N1b,` are the nodes tied to the near end and `N2a, N2b,…,NREF2` are the nodes tied to the far end. `WELFILENAME` is the name of the file that contains the R, L, G, and C data as a function of frequency. `LEN` is the length of the uniform coupled transmission lines with units using the MKSA system.

Examples:

```
W1 N=3 2 3 4 0 5 6 7 0 WEL LEN=2.95
```

Format of RLGC file

The RLGC file containing the data for the multiconductor system is in the following format

+ Number of lines (N)
+ Number of frequency points(M)
+ frequency_1 (f_1)
+ $R_{11}(f_1)$
+ $R_{12}(f_1)$
+ …
+ $R_{NN}(f_1)$
+ frequency_2(f_2)
+…
+ $R_{NN}(f_2)$
+…
+ frequency_M(f_M)
+…
+ $R_{NN}(f_M)$
+ frequency_1
+ $L_{11}(f_1)$
+ $L_{12}(f_1)$
+ …
+ $L_{NN}(f_1)$
+ frequency_2(f_2)
+…
+ $L_{NN}(f_2)$
+…
+ frequency_M(f_M)
+…
+ $L_{NN}(f_M)$
+ frequency_1(f_1)
+ $G_{11}(f_1)$
+ $G_{12}(f_1)$
+ …
+ $G_{NN}(f_1)$
+ frequency_2(f_2)
+…
+ $G_{NN}(f_2)$
+…
+ frequency_M(f_M)
+…
+ $G_{NN}(f_M)$
+ frequency_1(f_1)
+ $C_{11}(f_1)$
+ $C_{12}(f_1)$
+ …
+ $C_{NN}(f_1)$
+ frequency_2(f_2)
+…
+ $C_{NN}(f_2)$
+…
+ frequency_M(f_M)
+…
+ $C_{NN}(f_M)$

R_{ij}, L_{ij}, G_{ij} and C_{ij} are the resistance, inductance, conductance and capacitance matrix elements per unit length respectively.

15. N element

The macromodel element is used to represent circuit blocks that are described by their frequency-dependent network parameters. This version will only process S parameters.

General form:

```
NXYZ NPORTS FILETYPE FILENAME N1 N2 N3 N4 ... NP A=Aval P=Pval
O=Oval
```

`NPORTS` is the number of ports

`FILETYPE` is a string that indicates the file type and can take one of two values: `SPAR` or `POLERES`. When the file type is `SPAR`, `FILENAME` is the name of the S-parameter file. `N1, N2,...,NP` is the list of nodes tied to the ports of the macromodel. When the file type is `POLERES`, `FILENAME` is the name of a data file containing the poles and residues of the macromodel. `FILETYPE` can be omitted in which case `FILENAME` will be assumed to be a S-parameter file.

Next a set of switches are defined to set parameters associated with the model-order reduction (MOR) for the extraction of the poles and residues of the N-port

`A` is the automatic order of approximation switch. `A=0` indicates that the order will be fixed by the user with a default value of 32. `A=1` means that the order of approximation is calculated automatically by the MOR routine.

`O` is the order of approximation (`Oval`). It is ignored if the order is calculated automatically (`A=1`)

`P` is the passivity switch. `P=0` indicates that passivity is not enforced; `P=1` indicates passivity enforcement.

S is a flag that determines whether the step response of the n-port is produced. if `S` = 1, the routine will generate the step response and write it in a file *element_name*_stepresponse

C is a flag that determines if a circuit netlist is produced from the pole/residue extraction. if `C =` 1 the routine will generate a SPICE3 compatible netlist and will write it in a file titled *element_name*_ckt_list. Note that the produced netlist may not be compatible with the LIM simulator since it may contain negative circuit elements.

R is a flag that controls the report of the passivity assessment/enforcement process. If `R = 1`, the routine will report the passivity violation instances in the out_file log.

`N` is an integer that specifies the maximum number of trials in the passivity enforcement process. If enforcement is not achieved within `N` trials, the routine will report a failure in passivity enforcement and proceed.

Examples:

```
NMAC 2 SDATA/LOSSYTL.S2P 2 3 A=0 P=1 O=45 C=1 R=1 N=25
```

Format of S-Parameter file

The S parameter file containing the data for the macromodel and should be arranged in Touchstone format

The first line in the Touchstone file reads

`!XYZ   freq_unit   format   ports`

`XYZ` is a string indicating a name or the file.

`freq_unit` is the unit for the frequency values.

`format` is a mnemonic for the format for the scattering parameter values. Valid formats are:

- `RI`: real and imaginary
- `MA`: magnitude(linear) and phase(degrees)
- `DB`: magnitude (dB) and phase(degrees)

`ports` is the number of ports for the macromodel

All subsequent lines in the file that start with a "!" are treated as comment lines. The number of such lines is arbitrary. We then have a line in the format:

`# freq_unit  param  format  Ref  Value`

where `freq_unit` is the unit for the frequency values.

`param` is a letter indicating the type of parameter. In this version of LIM only "`S`" is allowed

`format` specifies the format in which the scattering parameters are listed according to the definitions given above.

`Ref` is a letter indicating the type of the reference

`Value` is the value of the reference system

After the headers, the data is listed on a line by line format where each line is associated with frequency:

```
frequency  param_11(a) param_11(b) param_12(a) param_12(b)  param_ij(a)
param_ij(b)
```

In this form, the `param_ij` refer to the network parameters associated with ports *i* and j of the macromodel. `(a)/(b)` refer to the respective component depending on the data format (eg real/imaginary or magnitude/phase) as defined on the first line.

Examples:

```
!_Coupled_Line Hz RI 4
!_Created_Thu_Feb_19_13:46:21_1998
#_hz_S_ri_R_50
```

```
!_4_port_Network_SP1.SP
!_Coupled_Line
1e+06 1.03389e-06  0.000629999     0.999999  -0.00112879  5.01921e-07  0.000327027 -4.43923e-07 -0.000210752
           0.999999  -0.00112879  1.03389e-06  0.000629999 -4.43923e-07 -0.000210752  5.01921e-07  0.000327027
         5.01921e-07  0.000327027 -4.43923e-07 -0.000210752  1.03389e-06  0.000629999     0.999999  -0.00112879
        -4.43923e-07 -0.000210752  5.01921e-07  0.000327027     0.999999  -0.00112879  1.03389e-06  0.000629999

   1.01e+08  0.00787306    0.0628062    0.990788    -0.113182   0.00504536    0.0323904  -0.00445385    -0.020664
           0.990788    -0.113182   0.00787306    0.0628062  -0.00445385    -0.020664   0.00504536    0.0323904
         0.00504536    0.0323904  -0.00445385    -0.020664   0.00787306    0.0628062    0.990788    -0.113182
        -0.00445385    -0.020664   0.00504536    0.0323904    0.990788    -0.113182   0.00787306    0.0628062
```

Format of pole/residue file

The pole/residue file containing the data for the macromodel is arranged in the following format

The first line in the file reads

`N   port   filename`

In which `N` is the number of ports

The second line in the file reads

`M poles`

In which `M` is the total number of poles.

The third line in the file reads:

`NC complex pairs`

In which `NC` is the number of pairs of complex poles.

The fourth line in the file is in the form:

`NR real poles`

In which NR is the number of real poles.

The fifth line reads

`FM maximum frequency`

In which FM is the maximum frequency.

The sixth line reads

`Reference impedance matrix.`

The next N2 lines will be the entries of the impedance matrix in ohms

The next M lines will have the poles listed as:

`i real imag`

In which `i` is the index of the pole, real and imag are the real and imaginary parts of the pole respectively.

The following line will read

```
residue i j
```

Indicating that the `M` complex residues of S_{ij} will be listed

The next `2XM` lines list the residues of S_{ij} as

```
i real imag
```

In which `i` is the index of the residue, real and imag are the real and imaginary parts of the residue respectively.

Examples:

```
2 port LOSSYTL.S2P_poles_and_residues
110 poles
 54 complex pairs
 2 real poles
5.00000e+09 maximum frequency
reference impedance matrix
5.00000e+01
0.00000e+00
0.00000e+00
5.00000e+01
 1  -6.8089e+08 3.1152e+10
 2  -6.8089e+08 -3.1152e+10
 3  -2.2215e+08 -3.0004e+10
 4  -2.2215e+08 3.0004e+10
residue   1   1
 1  1.6827e+08 2.7410e+07
 2  1.6827e+08 -2.7410e+07
 3  -5.4046e+06 1.4497e+07
 4  -5.4046e+06 -1.4497e+07
constant term  6.57230e-02
residue   1   2
 1  -4.9799e+07 4.2615e+08
 2  -4.9799e+07 -4.2615e+08
 3  -2.2806e+07 7.1233e+06
:
:
109  -6.29897e+08
110  2.50134e+06
constant term  1.59294e-01
```

The step-response file contains the time-domain step responses associated with each of the n-port S parameters of the blackbox. The time-domain data is arranged as follows:

```
Number of time points  Number of ports

Time   s11   s12   s13 …sNN
:
:
```

The step function used for excitation assumes a rise time that is a hundred times the time step used for the simulation ($t_r = 100 \times \Delta t$).

Example

```
 10000   4 1.12500e-12
0.00000e+00 0.00000e+00 0.00000e+00 0.00000e+00 0.00000e+00
1.12500e-13 0.00000e+00 0.00000e+00 0.00000e+00 0.00000e+00 :
:
:
1.14750e-11 -2.24088e-07 2.47525e-09 3.13229e-09 -3.54925e-09
1.15875e-11 -6.67276e-07 7.22603e-09 9.18498e-09 -1.04076e-08
1.17000e-11 -1.32458e-06 1.40526e-08 1.79462e-08 -2.03349e-08
```

16. IBIS Models - P element

The P element is used to represent ibis behavioral models.



General form:

```
PXYZ IBISFILE.IBS T E IO PU PD PC GC
```

`IBISFILE.IBS` is the name of the ibis file. This file follows the prescribed ibis format. The nodes are defined as follows:

`T`: trigger - the buffer state is controlled by the voltage between this terminal and ground.

`E`: enable - enabling and disabling the buffer is controlled by the voltage between this terminal.

`IO`: input/output - when io is disabled, it is the input terminal when IO is enabled, it is the output terminal

`PU`: pull-up - power supply terminal for the pullup (PU) device.

`PD`: pull-down - power supply terminal for the pulldown (PD) device.

`PC`: power clamp - power supply terminal for the power clamp (PC)

`GC`: ground clamp - power supply terminal for the ground clamp (GC)

Refer to the ibis forum (http://ibis.org) for information related to the ibis file format. The LIM simulator contains an IBIS parser that will read the specified IBIS file and extract all the relevant parameters needed for the analysis

Examples:

```
PIBIS K4T1G164QE.IBS 17 18 16 19 20 21 22
```

17. Diodes – D-Element



General form:

```
DXYZ N+ N- MNAME
```

N+ and N- are the positive and negative nodes, respectively. MNAME is the model name. The general equation for the diode model is given by

$$I_D = I_S\left(e^{\frac{qV_D}{kT}} - 1\right) - I_Z e^{A(V_b - B'V_D)} + \frac{V_D}{R_C}$$

DIODE MODEL PARAMETERS

	Param	Description	Unit	Default
1	IS	saturation current	A	10^{-14}
2	RS	ohmic resistance	W	0
3	N	emission coefficient	-	1
4	TT	transit-time	sec	0
5	CJO	zero-bias junction capacitance	F	0
6	VJ	junction potential	V	1
7	M	grading coefficient	-	0.5
8	EG	activation energy	eV	1.11
9	XTI	saturation-current temp. exp	-	3
10	KF	flicker noise coefficient	-	0
11	AF	flicker noise exponent	-	1
12	FC	coefficient for forward-bias	-	0.5
13	BV	reverse breakdown voltage	V	inf
14	IBV	current at breakdown voltage	A	10^{-3}
15	TNOM	parameter measurement temperature	C	27
16	RC	Zener resistance (R_C)		inf
17	A	Exponential factor (A)		1
18	BP	B attenuation constant (B')		1.0
19	VB	Zener voltage(V_b)	V	0
20	IZ	Zener saturation current (I_z)	A	0

MNAME is the name of the diode model which must be found in a .MODEL line. Such a line will have the form:

```
.MODEL MNAME D IS=Value N=1 XTI= Value RC=value A=1.0 BP=1.15
+ VB=-18.0 IZ=1.0
```

Examples:

```
DBRIDGE 2 10 DIODE1
DCLMP 3 7 DMOD 3.0 IC=0.2
```

18. Bipolar Junction Transistors (BJTs) – Q-Element



General form:

```
QXYZ NC NB NE NS MNAME
```

`NC`, `NB`, and `NE` are the collector, base, and emitter nodes, respectively. `NS` is the (optional) substrate node. If unspecified, ground is used. `MNAME` is the name of the BJT model which must be found in a `.MODEL` line. Such a line will have the form:

`.MODEL MNAME TYPE LEVEL=L PARAM=VALUE …`

`TYPE` can be "`NPN`" or "`PNP`"

`LEVEL` refers to the model used. Currently, two models are available which are specified by the index `L`. When `L` is 1 The Ebers-Moll model is used. When `L` is 2, the VBIC model is used. These two models are described in the paragraphs below.

Examples:

```
Q23 10 24 13 QMOD
Q50A 11 26 4 20 MOD1
```

BJT EBERS-MOLL MODEL PARAMETERS

	Param	**Description**	**Unit**	**Default**
1	BF	forward beta	-	100
2	BR	reverse beta	-	1
3	IS	saturation current	A	10^{-16}
4	VAF	forward Early voltage	V	infinite
5	VAR	reverse Early voltage	V	infinite
6	RB	zero bias base resistance	ohms	0
7	RC	collector resistance	ohms	0
8	RE	emitter resistance	ohms	0
9	CJE	base-emitter zero-bias depletion capacitance	F	$2x10^{-12}$
10	TR	reverse transit time	sec	0
11	TF	ideal forward transit time	sec	0
12	CJC	base-collector zero-bias depletion capacitance	F	0

VBIC - model

LIM also implements the VBIC (Vertical Bipolar Inter Company) model is a extended development of the Standard Gummel-Poon (SGP) model with the focus of integrated bipolar transistors. In LIM, the VBIC model is specified by setting LEVEL=2. A circuit schematic of the VBIC model is shown below.



VBIC Parameters

A partial list of VBIC parameters is listed below. For a complete list please refer to: http://synclesis.com/LIM/vbic

Name	Description	Unit	Default
Tnom	Parameter measurement temperature	°C	27.
Rcx	Extrinsic collector resistance	Ω	0.1
Rci	Intrinsic collector resistance	Ω	0.1
Vo	Epi drift saturation voltage	V	Infin.
gamm	Epi doping parameter		0.0
hrcf	High current RC factor		Infin.
Rbx	Extrinsic base resistance	Ω	0.1

Rbi	Intrinsic base resistance	Ω	0.1
Re	Intrinsic emitter resistance	Ω	0.1
Rs	Intrinsic substrate resistance	Ω	0.1
Rbp	Parasitic base resistance	Ω	0.1
Is	Transport saturation current	A	1e-16
nf	Forward emission coefficient		1.
nr	Reverse emission coefficient		1.
Fc	Fwd bias depletion capacitance limit		0.9
Cbeo	Extrinsic B-E overlap capacitance	F	0.0
Cje	Zero bias B-E depletion capacitance	F	0.0
pe	B-E built in potential	V	0.75
me	B-E junction grading coefficient		0.33
Aje	B-E capacitance smoothing factor		-0.5
Cbco	Extrinsic B-C overlap capacitance	F	0.
Cjc	Zero bias B-C depletion capacitance	F	0.
Qco	Epi charge parameter	C	0.
Cjep	B-C extrinsic zero bias capacitance	F	0.
pc	B-C built in potential	V	0.75
mc	B-C junction grading coefficient		0.33

References:

C. C. McAndrew et al., "Vertical Bipolar Inter Company 1995: An Improved Vertical, IC Bipolar Transistor Model", Proceedings of the IEEE Bipolar Circuits and Technology Meeting, pp. 170 - 177, 1995

C. C. McAndrew et.al., VBIC95, "The Vertical Bipolar Inter-Company Model", IEEE Journal of Solid State Circuits, vol. 31, No. 10, October 1996

C. C. McAndrew, VBIC Model Definition, Release 1.2, 18. Sep. 1999

19. MOSFETs



General form:

```
MXYZ ND NG NS NB MNAME L=VAL W=VAL
```

`ND, NG, NS`, and `NB` are the drain, gate, source, and bulk (substrate) nodes, respectively. `MNAME` is the model name. L and W are the channel length and width, in meters.

Example:

```
M1 24 2 0 20 TYPE1
M31 2 17 6 10 MODM L=5U W=2U
M1 2 9 3 0 MOD1 L=10U W=5U
```

Models for MOSFETS

The `.MODEL` statement for a MOSFET reads

```
.MODEL MODELNAME POLARITY LEVEL=VALUE P1=VAL P2=VAL ...PN=VAL
```

`.MODEL` where `MODELNAME` is the name of the model specified in the element line, `POLARITY` is either "`NMOS`" for N-channel devices or "`PMOS`" for P-channel devices. `LEVEL` specifies the level number for the model, next the `PJ=VAL` specify the values of some of the parameters associated with the model level.

MOSFET LEVEL 1 MODEL PARAMETERS

	Parameter	Description	Unit	Default
1	UO	Surface mobility	cm^2/V	600
2	UCRIT	Critical field for mobility degradation	V/cm	
3	UEXP	critical field exponent in mobility degradation -		
4	TOX	oxide thickness	meter	10^{-7}
5	XJ	metallurgical junction depth	meter	0
6	CGSO	gate-source overlap capacitance per meter channel width	F/m	0
7	CGDO	gate-drain overlap capacitance per meter channel width	F/m	0
8	CBD	zero-bias B-D junction capacitance	F	0
9	CBS	zero-bias B-S junction capacitance	F	0
10	LD	lateral diffusion	meter	0
11	NSS	surface state density	$1/cm^2$	0
12	KP	transconductance parameter	A/V^2	2.0e-5
13	PHI	surface potential	V	0.6
14	VTO	zero-bias threshold voltage (VTO)	V	0
15	DELTA	width effect on threshold voltage -	0	
16	GAMMA	bulk threshold parameter	$V^{1/2}$	0
17	LAMBDA	channel-length modulation	1/V	0
18	ETA	static feedback -	0.0	

19	KAPPA	saturation field factor	-	0.2
20	THETA	mobility modulation	1/V	0
21	MJ	bulk junction bottom grading coeff.	-	0.5
22	MJSW	bulk junction sidewall grading coeff.	-	0.5
23	CJ	zero-bias bulk junction bottom cap.per sq-meter of junction area	F/m^2	0
24	CJSW	zero-bias bulk junction sidewall cap. per meter of junction perimeter	F/m	0
25	PB	bulk junction potential	V	0.8

Level 1 is the most basic model employed by LIM. It is based on the Shichman-Hodges model. It is used mainly for long-channel devices. The charge-storage dynamic behavior for this levels employs the Meyer model, which is not a charge-conserved model.

LIM also supports more advanced levels based on the BSIM (Berkeley Short-Channel IGFET Model) project. These models employ empirical parameters and polynomial equations to handle various physical effects. This generally leads to improved circuit simulation behavior compared to earlier models. It becomes less accurate for although its accuracy degrades in submicron FETs. Currently available in LIM are BSIM3 (Level 37) and BSIM4 (Level 54) models. An extensive list of the BSIM3 and BSIM4 parameters and their default values can be found in the literature. BSIM models belong to the category of the third generation of MOSFET models that fulfilled two main purposes. The first is the need for models suitable for analog design. The second is to reduce the number of model parameters from that found in the second-generation models, and return more physical meaning to those parameters.

There are some limitations in the models. At this stage accuracy and convergence are not guaranteed for any combination of device parameters. Furthermore, the polynomial equations used in these models can behave poorly, causing negative electrical parameters and convergence problems. The improvement of the algorithms and models is an ongoing effort.

LIM also supports (in beta mode) the FinFET model based on the BSIM-CMG. Due to their excellent low power and scaling characteristics, FinFET devices have been adopted in all the MOS technology nodes below 20 nanometers. For this technology, the Compact Model Coalition (CMC) has selected BSIM-CMG as the first and only industry-standard compact model for advanced circuit design. BSIM-CMG compact model includes several physical effects: device geometry effects, charge quantization, gate oxide tunneling, gate capacitance degradation, short-channel effects. The table below shows the different models supported by LIM for various MOS technologies and their associated level number.

MOS MODELS AND LEVELS SUPPORTED BY LIM

Technology	**Model**	**Level**
>1000 nm	Shichman-Hodges	1
180 nm	BSIM3	37
130 nm	BSIM3	37

90 nm	BSIM4	54
65 nm	BSIM4	54
45 nm	BSIM4	54
32 nm	BSIM4	54
22 nm	BSIM4	54
FinFet (< 10 nm)	BSIM-CMG	57

MOSFET Parameters

A partial list of MOSFET parameters is listed below. For a complete list please refer to: http://synclesis.com/LIM/mosfets

20. Nonlinear Voltage-Controlled (Current or Voltage) Source – B-Element



General form:

```
BXYZ N+ N- VA=f(VB, VC, VD, ...)
BGHF N+ N- IA=f(VB, VC, VD, ...)
```

Example:

```
BEL1 4 0 V=V(3)-0.3*V(2)^3
```

N+ is the positive node, and N- is the negative node. The formula on the right-hand side of the equal sign expresses the value of the voltage or current as a nonlinear function of one or several controlling node voltages. The value resulting from the expression for the V and I variables are the voltages and currents across or through the nonlinear element, respectively. If I is used then the element is a current source, and if V is used the device is a voltage source. One and only one of these variables must be used.

For the expression, the following standard operators are defined: +, -, *, /, ^.

In addition, a list of transcendental functions of real variables are also available for use.These functions are: `abs`, `asinh`, `cosh`, `sin`, `acos`, `atan`, `exp`, `sinh`, `acosh`, `atanh`, `ln`, `sqrt`, `asin`, `cos`, `log`, `tan`.

21. Memristor – Z-Element



The memristor device has 2 terminals.

General form:

```
ZXXY N+ N- RON=VAL ROFF=VAL RINIT=VAL BETA=VAL VT=VAL B1=VAL
B2=VAL
```

`N+` and `N-` are the positive and negative nodes, respectively. `RON`, `ROFF` and `RINIT` are the values of the resistance in the on, off and initial states respectively. `BETA`, `VT`, `B1` and `B2` are other parameters that control the behavior of the memristor. The general equation for the memristor is given by

Example:

```
MEM1 1 2 Ron=1k Roff=10k Rinit=5k beta=1e13 b1=10u b2=10u
```

Model:



The memristor is equivalent to the current source G_{pm}. The voltage at the node x is denoted as V_x, which has initial value of R_{init}. V_x is calculated by the subcircuit with current source G_x, capacitor $C_x = 1$ pF, and a resistor $R_{aux} = 100$ MΩ in parallel.

$$G_{pm} = V_M/V_x,$$

$$V_M = V(N1) - V(N2),$$

$$G_x = f(V_m, b1) * W(V_x, V_M, b1, b2) \times 1p,$$

$$f(V_M) = \beta[V_M - 0.5(|V_m + V_t| - |V_m - V_t|)],$$

$$W(x, V_M, b1, b2) = \theta(V_M, b1)\theta\left(1 - \frac{x}{R_{off}}, b2\right) + \theta(-V_M, b1)\theta\left(\frac{x}{R_{on}} - 1, b2\right),$$

$$h(x, b) = x[\theta(x, b) - \theta(-x, b)],$$

$$\theta(x, b) = \frac{1}{1 + e^{-x/b}},$$

MEMRISTOR PARAMETERS

	Param	**Description**	**Unit**	**Default**
1	RON			
2	ROFF			
3	RINIT			
4	BETA			
5	VT			
6	B1			
7	B2			

Example 1 - NMOS Inverter – BSIM4

File: nmos_tsmc4.cir

```
*** NDINV * 4
MJOS 3 1 4 6 tsmc18dN L=1u W=12u
RS 4 0 2000
RD 2 3 10000
RG 1 5 50.0
RB 6 0 1.0
VDD 2 0 1.3
VP 5 0  PULSE (0 1.3  0.40000e-09  0.10000e-09  0.10000e-09  0.30000e-09  9.00000e-09)
.TRAN  3.12500e-13 1.30e-09
.PRINT TRAN V(1) V(2) V(3) V(4)
.PLOT TRAN V(1) V(2) V(3) V(4)
.LIM C=5.010e-15 L=0.01e-09 G=1.0e-20 P=200 D=0

.MODEL tsmc18dN NMOS LEVEL = 54
+VERSION = 3.1            TNOM    = 27             TOX     = 4.1E-9 TOXP = 20.0e-09
+DVT0    = 1.4728771   DVT1 = 5.0 DVT2    = 0.0213359
+XJ      = 1E-7           NCH     = 2.3549E17      VTH0    = 0.3647749
+U0 = 0.02656444712  UA      = -1.423009E-9   UB      = 2.335093E-18
+K1      = 0.5815814      K2      = 6.025001E-3    K3      = 1E-3
+K3B     = 1.4745568      W0      = 1E-7           NLX     = 1.632187E-7
+DVT0W   = 0              DVT1W   = 0              DVT2W   = 0
+UC      = 5.245012E-11   VSAT    = 9.412881E4     A0      = 1.8428561
+AGS     = 0.4038804      B0      = 5.524599E-7    B1      = 5E-6
+KETA    = -7.935044E-3   A1      = 0.8            A2      = 0.8804594
+RDSW    = 105            PRWG    = 0.4944853      PRWB    = -0.2
+WR      = 1              WINT    = 2.60605E-9     LINT    = 1.939129E-8
+XL      = 0              XW      = -1E-8          DWG     = -2.843075E-9
+DWB     = 2.865387E-9    VOFF    = -0.0894361     NFACTOR = 2.3051876
+CIT     = 0              CDSC    = 2.4E-4         CDSCD   = 0
+CDSCB   = 0              ETA0    = 2.398164E-3    ETAB    = 3.392679E-5
+DSUB    = 8.234246E-3    PCLM    = 0.7444877      PDIBLC1 = 0.185852
+PDIBLC2 = 3.274134E-3    PDIBLCB = -0.1           DROUT   = 0.7550884
+PSCBE1  = 8E10           PSCBE2  = 1.726969E-9    PVAG    = 1.068222E-3
+DELTA   = 0.01           RSH     = 6.8            MOBMOD  = 1
+PRT     = 0              UTE     = -1.5           KT1     = -0.11
+KT1L    = 0              KT2     = 0.022          UA1     = 4.31E-9
+UB1     = -7.61E-18      UC1     = -5.6E-11       AT      = 3.3E4
+WL      = 0              WLN     = 1              WW      = 0
+WWN     = 1              WWL     = 0              LL      = 0
+LLN     = 1              LW      = 0              LWN     = 1
+LWL     = 0              CAPMOD  = 2              XPART   = 0.5
+CGDO    = 8.28E-10       CGSO    = 8.28E-10       CGBO    = 1E-12
+CJ      = 9.427065E-4    PB      = 0.8            MJ      = 0.3709737
+CJSW    = 1.928292E-10   PBSW    = 0.7            MJSW    = 0.1902367
+CJSWG   = 3.3E-10        PBSWG   = 0.7            MJSWG   = 0.1902367
+CF      = 0              PVTH0   = 3.643308E-5    PRDSW   = -0.8968745
+PK2     = 3.285774E-4    WKETA   = -5.791315E-4   LKETA   = -0.0101627
+PU0     = 10.0548745     PUA     = 1.901573E-11   PUB     = 1.831424E-24
+PVSAT   = 1.433548E3     PETA0   = 5.358064E-5    PKETA   = 1.519426E-3
+NOIMOD  = 1              KF      = 4.5E-29      AF      = 1
+EF      = 1
.END
```


2
V_DD = 5 V
+
-
R_D=1000 Ω
3
R_G = 50 Ω
5
1
MJOS (N342)
4
V_P
R_S = 2000 Ω


V(1)
V(2)
V(3)
V(4)
Voltage(V)
Time(sec)
1.6
1.4
1.2
1
0.8
0.6
0.4
0.2
0
-0.2
0
0.2
0.4
0.6
0.8
1
1.2
1.4

Example 2 - Diode Clipper

File: clipper.cir

```
*main circuit
*vin 1 0  pulse(    0.00000    12.00000     1.00000n     1.00000n     1.20000n
12.00000n   91.20000n )
vin 1 0 sin (0 12.0 41.66e+06 2.0e-09 0.0 )
rg1 1 2 8000.0
d1 2 3 diode
d2 4 2 diode
vdc1 3 0 3.0
vdc2 0 4 1.0
rg2 3 0 500.0
.tran 0.01n   45.60000n
.print tran  v(1) v(2) v(3) v(4)
.plot tran  v(1) v(2) v(3) v(4)
.MODEL diode D IS=2.5e-14 N=0.5
.end
```





File: nand47.cir

```
* cmos nand
M1 5 1 0 0 N342 L=5U W=5U
M2 3 2 5 5 N342 L=5U W=5U
M3 3 1 4 4 P342 L=5U W=5U
M4 3 2 4 4 P342 L=5U W=5U
C1 5 0 1e-14
C2 3 0 1e-14
VA1 2 0 PWL (0 0 5e-08 0 6e-08 6 11e-08 6 12e-08 0 15e-08 0)
VB1 1 0 PWL (0 0 2e-08 0 3e-08 6 5e-08 6 6e-08 0 8e-08 0 9e-08 6 11e-08 6 12e-08 0 15e-08
0)
VVDD 4 0 6

.TRAN 1e-12 1.4e-07
.PRINT TRAN V(1) V(2) V(3) V(4)
.PLOT TRAN V(1) V(2) V(3) V(4)
*** 0.5um CMOS technology
*SPICE LEVEL3 PARAMETERS
.MODEL N342 NMOS LEVEL=3 PHI=0.7 TOX=1.98E-08 XJ=0.2U TPG=1
+ VTO=0.7 DELTA=8.8E-01 LD=5E-08 KP=1.56E-04
+ UO=420 THETA=2.3E-01 RSH=2.0E+00 GAMMA=0.62
+ NSUB=1.40E+17 NFS=7.20E+11 VMAX=1.8E+05 ETA=2.125E-02
+ KAPPA=1E-01 CGDO=3.0E-10 CGSO=3.0E-10
+ CGBO=4.5E-10 CJ=5.50E-04 MJ=0.6 CJSW=3E-10
+ MJSW=0.35 PB=1.1

*** 0.5um CMOS technology
*SPICE LEVEL3 PARAMETERS
.MODEL P342 PMOS LEVEL=3 PHI=0.7 TOX=1.98E-08 XJ=0.2U TPG=-1
+ VTO=-0.95 DELTA=2.5E-01 LD=7E-08 KP=4.8E-05
+ UO=130 THETA=2.0E-01 RSH=2.5E+00 GAMMA=0.52
+ NSUB=1.0E+17 NFS=6.50E+11 VMAX=3.0E+05 ETA=2.5E-02
+ KAPPA=8.0E+00 CGDO=3.5E-10 CGSO=3.5E-10
+ CGBO=4.5E-10 CJ=9.50E-04 MJ=0.5 CJSW=2E-10
+ MJSW=0.25 PB=1
.END
```





Example 4 - CMOS Inverter

File cmosinv.cir

```
*** CMOSINV * 4
MJOS 4 2 0 0 N342 L=1.2U W=5U
MSCH 4 2 3 3 P342 L=1.2U W=15U
RG 1 2 50
RD 4 0 10000
VDD 3 0 5
CDD 3 0 1.E-06
VP 1 0  PULSE (0 5  0.10000e-09  0.10000e-09  0.10000e-09  2.00000e-09  6.00000e-09)
.TRAN  1.12500e-12 4.00e-09
.PRINT TRAN V(1) V(2) V(3) V(4)
.PLOT TRAN V(1) V(2) V(3) V(4)

**** LEVEL 1 NMOS ****
.MODEL N342 NMOS
+ LEVEL=1 TPG=1
+ KP=3.6E-05
+ LAMBDA=0.00033 VT0=0.69486 GAMMA=0.60309 PHI=1
+ TOX=1.9800000E-08  XJ=0.2U  LD=0.1U  NSUB=4.9999999E+16
+ NSS=0.0000000E+00
+ CJ=4.091E-4   MJ=0.307   PB=1.0
+ CJSW=3.078E-10     MJSW=1.0E-2
+ CGSO=3.93E-10      CGDO=3.93E-10

**** LEVEL 1 PMOS ****
.MODEL P342 PMOS
+ LEVEL=1 TPG=-1
+ KP=7.69968E-06
+ LAMBDA=0.00033 VT0=-0.69486 GAMMA=0.89213 PHI=1
+ TOX=1.9800000E-08  XJ=0.2U  LD=0.1U   NSUB=4.9999999E+16
+ NSS=0.0000000E+00
+ CJ=6.852E-4        MJ=0.429           PB=1.0
+ CJSW=5.217E-10     MJSW=0.351
+ CGSO=7.29E-10      CGDO=7.29E-10
.END
```





SPICE vs LIM Elements

Element	SPICE	LIM	Comment
A	Convolution Macromodel	Macromodel-convolution	
B	Nonlinear dependent source	Nonlinear dependent source	
C	Capacitor	Capacitor	
D	Diode	Diode	
E	Voltage-dependent voltage src	Voltage-dependent voltage src	
F	Current-dependent current src	Current-dependent current src	
G	Voltage-dependent current src	Voltage-dependent current src	
H	Current-dependent voltage src	Current-dependent voltage src	
I	Independent current source	Independent current source	
J	JFET	**NOT YET JFET**	
K	Mutual inductance	Mutual inductance	
L	Inductance	Inductance	
M	MOSFETS	MOSFETS	
N	Macromodel	Macromodel-MOR	
O	Lossy TL	Lossy Transmission Line	
P	IBIS Model	IBIS	
Q	BJT	BJT	
R	Resistor	Resistor	
S	Switches	**NOT YET Switches**	
T	Lossless Transmission Line	Lossless Transmission Line	
U	Lossless Multiconductor	Multi-Coupled TL	
V	Independent Voltage Source	Independent Voltage Source	
W	W-Element (HSPICE)	W-Element (F-dep MTL)	
X	Subcircuit	Subcircuit	
Y	**NOT USED**	Nonlinear Element (IV Table)	
Z	MEMRISTOR	Memristor	